# Maintainer: Jaroslav Lichtblau <svetlemodry@archlinux.org>
# Maintainer: Alexander Rødseth <rodseth@gmail.com>
# Contributor: Fabian Bornschein <fabiscafe-at-mailbox-dot-org>
# Contributor: Hugo Doria <hugo@archlinux.org>
# Contributor: Christer Solskogen (solskogen@carebears.mine.nu)

pkgname=gtranslator
pkgver=45.3
pkgrel=1
pkgdesc='An enhanced gettext po file editor for the GNOME desktop environment'
arch=('loong64' 'x86_64')
url='https://projects.gnome.org/gtranslator/'
license=('GPL3')
depends=(
  'gettext'
  'gtksourceview5'
  'hicolor-icon-theme'
  'iso-codes'
  'libadwaita'
  'libgda6'
  'libspelling'
)
makedepends=(
  'gobject-introspection'
  'meson'
  'yelp-tools'
)
source=(https://download.gnome.org/sources/$pkgname/${pkgver%.*}/$pkgname-$pkgver.tar.xz)
sha256sums=('3010204df5c7a5ae027f5a30b1544d6977d417f0e4bb9de297f0ad1a80331873')

build() {
  arch-meson $pkgname-$pkgver build
  meson compile -C build
}

check() {
  meson test -C build --print-errorlogs
}

package() {
  meson install -C build --destdir "$pkgdir"
}
